Lyceum

Aristotle's Lyceum is the institution considered to be the forerunner of the modern university. Opened in 335 BC, the Lyceum was a center of study and research in both science and philosophy.

Extending the FD Method

Dividends

Last issue we saw how to solve the Black-Scholes equation using the explicit finite-difference method. This issue we'll see how to modify the previous code to include a dividend yield on the stock and to price American options. When the underlying asset is a currency with a known foreign interest rate or is a stock with dividends then the BS equation becomes

$$\Theta + \frac{1}{2}\sigma^2 S^2 \Gamma + (r - D)S\Delta - rV = 0\,.$$

Here D is the dividend yield on the stock or the foreign interest rate.

Early exercise

American options may be exercised at any time, not just expiry. If we believe that arbitrage opportunities shouldn't exist then we have to make sure that the theoretical option value is at least the payoff. Mathematically, we want

$$V \geq \text{Payoff}\,(S)\,.$$

where 'Payoff' is the payoff function. We also want the delta to be continuous, we don't want any jumps in the hedge ratio. This is incredibly easy to implement in the explicit finite-difference method, as you'll see below.

The explicit finite-difference code

Below is the Visual Basic code for pricing an American call option on a dividend-paying stock. The notation is the same as last time, with the addition of `Divvie` as the dividend yield on the underlying, `US` determines whether the option is American (`US = "Y"`) or not, and the `Payoff(i)`.

```
Function USOptionValue(Asset As Double, Strike As Double, Expiry As Double, _
    Volatility As Double, Divvie as Double, Intrate As Double, _
    US As String, NoAssetSteps As Integer) '(a)
Dim S(0 To 100), VOld(0 To 100), VNew(0 To 100) As Double
Dim Payoff(0 To 100) As Double '(b)
dS = 2 * Strike / NoAssetSteps
NearestGridPt = Int(Asset / dS)
dummy = (Asset - NearestGridPt * dS) / dS
Timestep = dS * dS / Volatility / Volatility / (4 * Strike * Strike)
NoTimesteps = Int(Expiry / Timestep) + 1
Timestep = Expiry / NoTimesteps
For i = 0 To NoAssetSteps
S(i) = i * dS
VOld(i) = Application.Max(S(i) - Strike, 0)
Payoff(I) = Vold(i) '(c)
Next i
For j = 1 To NoTimesteps
For i = 1 To NoAssetSteps
Delta = (VOld(i + 1) - VOld(i - 1)) / (2 * dS)
Gamma = (VOld(i + 1) - 2 * VOld(i) + VOld(i - 1)) / (dS * dS)
VNew(i) = VOld(i) + Timestep * (0.5 * Volatility * Volatility * S(i) * S(i) _
    * Gamma + (Intrate – Divvie) * S(i) * Delta - Intrate * VOld(i)) '(d)
Next i
VNew(0) = 0
VNew(NoAssetSteps) = 2 * VNew(NoAssetSteps - 1) - VNew(NoAssetSteps - 2)
For i = 0 To NoAssetSteps
VOld(i) = VNew(i)
Next i
If US = "Y" Then
For i = 0 To NoAssetSteps
VOld(i) = Application.Max(Vold(i), Payoff(i)) '(e)
Next i
End If
Next j
USOptionValue = (1 - dummy) * VOld(NearestGridPt) + _
    dummy * VOld(NearestGridPt + 1)
End Function
```

Notes:

(a) The function call now asks for dividend information and whether the option is American or not

(b) We must remember the payoff

(c) Setting up the payoff

(d) This includes the dividend yield

(e) This is the line that tests whether the option should be exercised

Exercise:

Change the above code to value Bermudan options, options for which exercise is permitted on or between specified dates.